Beam Dynamics with MAD

Roger Barlow

Lecture 1

What is MAD?

MAD: Methodical Accelerator Design
Well established language - now on version 10 (hence MADX)
Enables you to
(1) Describe the properties of an accelerator, or a beam line.
(2) Track one or many particles through the accelerator
(3) Adjust the design parameters to give desired properties ("matching")
(4) Plot the results

Getting going...

- Download MADX onto your laptop/desktop from `http://mad.web.cern.ch/mad/`. Details depend on your platform (Windows, linux, Mac...) but should not be a problem.
  Or log in to iiaa1.hud.ac.uk and use `/opt/madx`
- Start the program by clicking on the icon, or typing 'madx', as appropriate. Type commands. For example
  `print, text="Hello World!";`
- For longer problems, prepare input file (say, `myfile.txt`) with your favourite editor (not MSWord!), invoke MAD and and type
  `CALL, FILE=myfile.txt;`.
  Or (linux) type `madx < myfile.txt`
  Or (Windows) drag and drop `my file.txt` onto the desktop MADX icon. But you then lose valuable diagnostic output. Not recommended.
- Typical file contains (1) description of accelerator and (2) actions relating to it

Accelerator design

Reference Orbit There must be a stable closed path for a particle

Focussing Particles with trajectories near the reference orbit must oscillate about it, not drift away

Acceleration Particles must arrive at right point in RF cycle. Save for later lecture

Doing stuff Extraction, injection, collision, radiation. Save for later lecure

Co-ordinate system

Specify position *relative to the reference orbit* with (x, y, s)
s is the distance along the reference orbit. s and t essentially the same - apart from some constant velocity factor.
y is vertical
x is horizontal

Cyclic accelerators

Dipole Magnets

Charged particles bend in magnetic field: $\vec{F} = q\vec{v} \times \vec{B}$







Key equation

$$p = 0.3BR$$

Units: GeV/c, Tesla, metres. The 0.3 is c and GeV factor of 10^9

Note - relativistically correct

Focussing the beam

Quadrupole Magnets



Vertical field proportional to horizontal displacement (and vice versa)

$B_y \propto x \qquad B_x \propto y$

Two types:

F: Focussing in horizontal (x) plane: Defocussing in vertical (y) plane

D: Defocussing in horizontal (x) plane: Focussing in vertical (y) plane

Amazing fact: F D doublet has net converging effect in both planes!



Magnets in MAD - a minimal first program

```
MYMACHINE: SEQUENCE,L=100;
D1:  SBEND,L=1.5,AT=1,ANGLE=PI/2;
Q1:  QUADRUPOLE,K1=0.23,AT=11,L=1;
Q2:  QUADRUPOLE,K1=-0.14,AT=16,L=1;
D2:  SBEND,L=1.5,AT=26,ANGLE=PI/2;
Q3:  QUADRUPOLE,K1=0.23,AT=36,L=1;
Q4:  QUADRUPOLE,K1=-0.14,AT=41,L=1;
D3:  SBEND,L=1.5,AT=51,ANGLE=PI/2;
Q5:  QUADRUPOLE,K1=0.23,AT=61,L=1;
Q6:  QUADRUPOLE,K1=-0.14,AT=66,L=1;
D4:  SBEND,L=1.5,AT=76,ANGLE=PI/2;
Q7:  QUADRUPOLE,K1=0.23,AT=86,L=1;
Q8:  QUADRUPOLE,K1=-0.14,AT=91,L=1;
ENDSEQUENCE;
BEAM ENERGY=10;
USE SEQUENCE=MYMACHINE;
TWISS;
PLOT HAXIS=S,VAXIS=BETX,BETY;
```

Notes on the MAD input language

Note

1. Upper/lower case doesn't matter
2. Use colon to give names. Helps if you make names unique
3. Sequence can contain many different types of element - here only two
4. You don't need to specify drifts. It puts them in for you.
5. SBEND is 'sector bend' as distinct from RBEND
6. element positions (`AT...`) must be in sequence
7. Element type followed by parameters. Some general, some specific to element type
8. Semicolons at the end of each statement

Improving the program

This is not an example of good coding.
In particular, define variables rather than repeating numbers. e.g.

```
LB=1.5; // Bending magnets 1.5 m long
NSECTORS=4;
AA=2*PI/NSECTORS;
MYMACHINE: SEQUENCE,L=100;
D1:  SBEND,L=LB,AT=1,ANGLE=AA;
...
```

Assignment can be immediate, when sequence is defined (`L = LB`)
or deferred, when sequence is used (`L := LB;`)
See also: subclasses, WHILE, nested sequences and macros

What does it do?

16 lines of description
1 line of acton (`TWISS`)
1 line about showing results (`PLOT`)
Plot created by last line in program in file madx.ps



Notice axes, two curves for β_x and β_y, and nifty magnet layout along top

Introducing β

The equation of motion for a particle can be written

$$x'' = -k(s)x \qquad \text{Hill's Equation}$$

y is similar. (Remember: x, y are displacements about the regular orbit.)
$\frac{d}{ds}$ is the same as $\frac{d}{dt}$ apart from a constant factor (the velocity)
We know how to solve $x'' = -kx$. Simple harmonic motion.

$$x = A\cos(\sqrt{k}s + \phi)$$

The general oscillation (called betatron oscillations) can be transformed to SHM: the vertical scale is the square root of the horizontal scale

$$x(s) = A\sqrt{\beta(s)}cos(\psi(s) + \phi) \qquad \frac{d\psi}{ds} = \frac{1}{\beta}$$

So the question about a particle 'What is $x(s)$?" becomes 'What is $\beta(s)$?'

Phase and Tune

The story of β continues

β maps actual trajectory between two points in the ring onto notional sine wave (SHM solution)

The phase difference between two points is the same for all particles: property of the lattice.

Phase difference for a complete circuit $= 2\pi Q$

Q: number of betatron oscillations in a complete orbit, called the tune

(Same for all particles. Though different for horizontal and vertical)

Given by output from TWISS command (look for 'Q1' and 'Q2')

Thinking about β



β scales the displacement and the timescale: expanded time by factor β means expanded displacement by factor $\sqrt{\beta}$

Another of thinking about β

β describes the beam envelope.
A bunch of particles will include all phases - large x, small x'; large x', small x, and everything in between
The size of a bunch depends on the distribution in amplitude A. But as it travels through the accelerator, the physical size will get larger at high β, small at low β.
So high β regions have large spread in x, small spread in x'. suitable for collimators, etc. low β regions have small spread in x (large in x'), suitable for collisions.

β and its friends

The Courant-Snyder or Twiss parameters

Take some point on the ring. Particle goes past many times. Different position x and direction x' each time. For SHM, (x, x') points fall on rectangular ellipse (squashed circle). This is not quite SHM: they fall on a general ellipse centred on the origin

$\gamma x^2 + 2\alpha x x' + \beta x'^2 = \epsilon$ ϵ describes the size: area is $\pi\epsilon$.



α, β, γ describe the orientation and aspect ratio of the ellipse. It's the same β One redundant parameter: $\gamma = \frac{1+\alpha^2}{\beta}$. And $\alpha = -\frac{1}{2}\frac{d\beta}{ds}$

x, x' plot (or equivalent x, p_x plot). is *phase space*

At different points on the lattice there are different ellipses - different α, β, γ - but ϵ stays constant: Liouville's theorem says the area is the same.

Proof that this is the same β

At some point s, displacement is $x = A\sqrt{\beta}\cos(\psi+\phi)$

Reminder: A and ϕ depend on the particle, but are constant. β and ψ depend on s but are the same for all particles.

Differentiate: $x' = \frac{A\beta'}{2\sqrt{\beta}}\cos(\psi+\phi) - A\psi'\sqrt{\beta}\sin(\psi+\phi) =$
$\frac{-A\alpha}{\sqrt{\beta}}\cos(\psi+\phi) - \frac{A}{\sqrt{\beta}}\sin(\psi+\phi)$

Rearrange: $\cos(\psi+\phi) = \frac{x}{A\sqrt{\beta}}$ $\qquad$ $\sin(\psi+\phi) = -(x' + \frac{\alpha}{\beta}x)\frac{\sqrt{\beta}}{A}$

Invoke $\sin^2+\cos^2 = 1$ and get $\beta x'^2 + 2\alpha x x' + (\frac{\alpha^2}{\beta} + \frac{1}{\beta})x^2 = A^2$

QED. (With $A^2 \equiv \epsilon$)

TWISS computes β. And other stuff...

How does MAD do it? You don't actually need to know, but...

For a drift of length L, $\begin{pmatrix} x \\ x' \end{pmatrix} \rightarrow \begin{pmatrix} x + Lx' \\ x' \end{pmatrix} = \begin{pmatrix} 1 & L \\ 0 & 1 \end{pmatrix} \begin{pmatrix} x \\ x' \end{pmatrix}$

For a quadrupole of strength K, $\begin{pmatrix} x \\ x' \end{pmatrix} \rightarrow \begin{pmatrix} x \\ x' - Kx \end{pmatrix} = \begin{pmatrix} 1 & 0 \\ -K & 1 \end{pmatrix} \begin{pmatrix} x \\ x' \end{pmatrix}$

So drifts and quads are represented by 2x2 matrices, often called **R**.
A succession of n drifts and quads (and...) can be written as a product of 2x2 matrices - also a 2x2 matrix. $\mathbf{R} = \mathbf{R_n R_{n-1} ... R_2 R_1}$. The matrix for a complete turn from some chosen starting point, $\mathbf{R_T}$, is of particular significance.

Putting $\begin{pmatrix} x \\ x' \end{pmatrix} \rightarrow \mathbf{R} \begin{pmatrix} x \\ x' \end{pmatrix}$ into $\gamma x^2 + 2\alpha x x' + \beta x'^2$, $\begin{pmatrix} \alpha \\ \beta \\ \gamma \end{pmatrix} \rightarrow \mathbf{M} \begin{pmatrix} \alpha \\ \beta \\ \gamma \end{pmatrix}$

where the 9 elements of **M** are made from the 4 elements of **R**. And $\mathbf{M_T}$ leaves α, β, γ unchanged. So they can be found at the starting point, as an eigenvector of $\mathbf{M_T}$, and thus at every point using the relevant **M**.

Just to show....

The 2x2 matrix transformation means

$$x \to R_{11}x + R_{12}x' \qquad x' \to R_{21}x + R_{22}x'$$

That takes $\gamma x^2 + 2\alpha xx' + \beta x'^2$ to

$$\gamma(R_{11}x + R_{12}x')^2 + 2\alpha(R_{11}x + R_{12}x')(R_{21}x + R_{22}x') + \beta(R_{21}x + R_{22}x')^2$$

$$= (\gamma R_{11}^2 + 2\alpha R_{11}R_{21} + \beta R_{21}^2)x^2$$
$$+(2\gamma R_{11}R_{12} + 2\alpha(R_{11}R_{22} + R_{12}R_{21}) + 2\beta R_{21}R_{22})xx'$$
$$+(\gamma R_{12}^2 + 2\alpha R_{12}R_{22} + \beta R_{22}^2)x'^2$$

$$\mathbf{M} = \begin{pmatrix} R_{11}R_{22} + R_{12}R_{21} & R_{21}R_{22} & R_{11}R_{12} \\ 2R_{12}R_{22} & R_{22}^2 & R_{12}^2 \\ 2R_{11}R_{21} & R_{21}^2 & R_{11}^2 \end{pmatrix}$$

Matching

Adjust one or more parameters to achieve target values of one or more properties

In the example, parametrise K1 with (say) KF, using deferred assignment:

```
q1:  QUADRUPOLE,K1 := KF,L=LQ,AT=...
```

Then add

```
MATCH;
VARY,NAME=KF,STEP=0.0001;// parameter to adjust
GLOBAL,Q1=2.3; // Goal is to get horiz.  tune equal to 2.3
MIGRAD,CALLS=20,TOLERANCE=1.E-20;
ENDMATCH;
```

Try it - with different tolerance, step size, etc

To achieve a non-global quantity, need to specify the place. Replace GLOBAL... line by something like

```
CONSTRAINT,RANGE=Q1,BETX=180;
```

Can have several variables/constraints. Note useful element MARKER

Tracking




```
TRACK,DUMP; // need 'DUMP' to save results
  START,PX=1.E-2; // initial offset
  RUN,TURNS=1000;
ENDTRACK;
PLOT,TABLE=TRACK,PARTICLE="1",FILE=trkplot,HAXIS=X,VAXIS=PX;
```

Note: for PLOT you will lead to download the gnu_plot program if you don't already have it

Summary

You have learned about:

1. The MAD program and how to run it
2. `SEQUENCE...ENDSEQUENCE`
3. `RBEND` and `SBEND`
4. `QUADRUPOLE`
5. `TWISS`
6. `MATCH`
7. `TRACK`
8. `PLOT`

Assignment

Use MADX to design a viable 20 GeV electron storage ring with 10-fold symmetry, with a quadrupole pair in each of the 10 sectors.
The circumference will be given at the lecture
Plot the beta functions around the ring.
Adjust the quadrupole strengths using MATCH to give a horizontal tune of 3.2, and a value of beta-y of 11.3 at the point midway between the two quadrupoles. (Vary quad positions by hand if necessary.)
Plot the beta functions over one of the 10 sectors.
Use TRACK and PLOT to show the vertical displacement phase space ellipse at the point midway between the quadrupoles.
Your answer should be submitted on UniLearn as a single document including your MAD commands, selected output, the requested plots, and the necessary words of explanation. I will expect to be able to run your MAD commands.